

**MESURES D'ORNIÉRAGE A L'AIDE
D'UN TRANSVERSOPROFILOGRAPHE**

CANQ
TR
GE
SM
133

**MESURES D'ORNIÉRAGE A L'AIDE
D'UN TRANSVERSOPROFILOGRAPHE**

MINISTÈRE DES TRANSPORTS
CENTRE DE DOCUMENTATION
700, boul. RENÉ-LÉVESQUE EST, 21^e étage
QUÉBEC (QUÉBEC) CANADA
G1R 5H1

TABLE DES MATIERES

1.	INTRODUCTION	1
2.	TRANSVERSOPROFILOGRAPHE	3
3.	AMELIORATION A APPORTER A L'APPAREILLAGE	4
4.	CONCLUSION	5

ANNEXES

- ANNEXE 1 : PHOTOS DE L'APPAREILLAGE
 - ANNEXE 2 : DESSIN DE L'APPAREILLAGE
-

1. INTRODUCTION

Depuis quelques années, le problème prédominant concernant les revêtements bitumineux apparaît être causé par les charges des poids lourds. Les charges augmentent sans cesse et les pneus radiaux à hautes pressions causent un déplacement du mélange bitumineux par période chaude d'été. Le Laboratoire Central effectue présentement diverses études afin d'augmenter la résistance à l'orniérage des revêtements.

Toutes ces études impliquent la mesure des ornières sur la route. A ce jour, seule l'utilisation d'une règle rigide de 1800 mm fut utilisée au Québec. Posée horizontalement, l'espace entre le bas de la règle et le fond de l'ornière est mesurée. Cette valeur nous donne la profondeur maximale de l'ornière.

La méthode utilisée est économique, mais présente le défaut majeur de ne pas indiquer la largeur de l'ornière ainsi que son type (simple, double, etc.). Pour palier à cet inconvénient, le Laboratoire Central a entrepris la construction d'un appareil qui trace à l'échelle le profil transversal de la surface d'une chaussée, Cet appareil est désigné sous le nom de transversoprofilographe.

ESSAI No. AUT7302

GAUCHE

DROITE

ROUTE.

TRONCON.

SECTION.

DATE.


CHAINAGE.

REGLE. 1800 mm

DEVERS. 1.21/100

PROF. 24.333
EAU. 25.033
A-B. 07233
AIRE. 1422733²

PROF. 22.033
EAU. 22.033
C-D. 07633
AIRE. 1218433²



MESURES D'ORNIERES

2. TRANSVERSOPROFILOGRAPHE

Cet appareil se compose de deux parties, soit:

- a) Appareil de prise des lectures de la surface de la chaussée (photo 1).
- B) Système d'acquisitions de données constitué d'un voltmètre programmable, ordinateur et traceur de courbe (photo 2).

Egalement s'ajoute une génératrice à courant stabilisé et divers accessoires secondaires.

Tout cet appareillage permet de:

- A) Tracer à l'échelle le profil transversal de la surface de la chaussée (figure 1).
- B) Calculer les mesures suivantes qui sont inscrites sur le graphique du profil transversal (figure 1):
 - profondeur de l'ornière
 - profondeur d'eau dans l'ornière
 - largeur de l'ornière
 - surface sous la droite de référence (A-B et C-B) ayant pour point d'appui la largeur de l'ornière
 - le devers de la route.

En annexe, l'on trouvera des photos de l'appareillage utilisé ainsi que des schémas. Le principe de prises de lecture est le suivant:

Un chariot se déplaçant horizontalement à une vitesse constante et est muni d'un lecteur de déplacement vertical. le lecteur est branché à un système d'acquisition de données et les coordonnées de déplacement vertical du lecteur de déplacement sont enregistrées. Les coordonnées du déplacement horizontal sont calculées à partir du temps que prend le chariot pour parcourir la distance entre le point de référence au départ et le point à son arrivée ainsi que du nombre de lectures prises par le lecteur de déplacement vertical.



3. AMÉLIORATION A APPORTER A L'APPAREILLAGE

L'essai est précis, mais l'installation de la règle sur laquelle se déplace le chariot est trop long à notre avis, soit 3 à 4 minutes. Un angulomètre digital permettrait d'installer la règle sans se soucier de son horizontalité. Egaleme nt une règle plus rigide serait la bienvenue et permettrait de ne pas fixer un support central temporaire pour l'essai.

4. CONCLUSION

Avec de légères modifications, le transversoprofilographe serait très efficace et nous croyons que le Ministère devrait l'utiliser sur toute section soumise à des expertises ou études concernant l'orniérage.

ANNEXE 1
PHOTOS DE L'APPAREILLAGE





1	2
3	

Photo No 1 : Transversoprofilographe

Photo No 2 : Système d'acquisition

Photo No 3 : Moteur d'entraînement du chariot



4	5
6	

Photo No 4: Chariot et lecteur de déplacement vertical

Photo No 5: Véhicule et génératrice

Photo No6: Essais en cours

ANNEXE 2
DESSIN DE L'APPAREILLAGE

☐ TRANSVERSOPROFILOGRAPHIE





TABLE DES PLANS


- ☐ SYSTEME DE ROULEMENT p.1, p.2, p.4, p.5
- ☐ ARBRES p.3
- ☐ SYSTEME DE LECTURES p.6
- ☐ SYSTEME D'ENTRAÎNEMENT p.7, p.8

□ ARBRES (EN ACIER)




ARBRE	QUANTITE	DIMENSIONS			
		A	B	C	D
#1	4	3	28	8	15
#2	4	3	49	8	15
#3	2	3	36	8	15

SYSTEME DE ROULEMENTS




ROULEMENT

TYPE: FAGIR LRCR 3/4






PLAN DU ROULEMENT




COUPE A-A


SYSTEME DE ROULEMENT




□ SYSTEME DE ROULEMENT (CHARIOT)




☐ SYSTEME DE ROULEMENTS





☐ PLAQUE AJUSTABLE (ALUMINIUM)




□ SYSTEME DE
LECTURES





☐ SYSTEME POUR RETENIR
 LE MOTEUR .

☐ SYSTEME D'ENTRAÎNEMENT
(TRANSMISSION PAR CABLE)



ANNEXE 3
PROGRAMMES D'INFORMATIQUE

- 1. ACQUISITION**
- 2. TRAITEMENT**

```

10 !      88/09/28
20 ! -----
30 ! NOM DU PROGRAMME: ACQORN
40 ! -----
50 !
60 ! PROGRAMME D'ACQUISITION POUR MESURES D'ORNIERES
70 !
80 ! SET-UP
90 !
100 CLEAR @ GCLEAR
110 !
120 ! DIMENSION DES VARIABLE
130 !
140 DIM X(500),Y(500),V(500)
150 PRINTER IS 2
160 CRT IS 1
170 NORMAL
180 MAT X=(0)@ MAT Y=(0)@ MAT V=(0)
190 !
200 ! ACQUISITION DES VALEURS
210 ! LES PLUS IMPORTANTE POUR
220 ! LE TRAITEMENT
230 !
240 DISP "NUMERO D'ESSAI"
250 DISP "(6 CAR.MAX.)"
260 INPUT F$
270 C1=8
280 K1=.2778
290 !
300 ! SAISIS DU VOLTAGE
310 !
320 OUTPUT 722 ; "PRESET"
330 OUTPUT 722 ; "TERM REAR"
340 OUTPUT 722 ; "DCV 40,.1"
350 OUTPUT 722 ; "CHAN 0"
360 ENTER 722 ; S1
370 DISP "LONGEUR POUR ESSAI EN 'MM'"
380 INPUT D1
390 CLEAR
400 DISP "VOUS ETES MAINTENANT PRET POUR PRENDRE LES LECTURES."
410 DISP
420 DISP "DEMARREZ LE CHARIOT ET FAIRE 'CONT' AU MOMENT OU LA ROUE"
430 DISP "PASSERA SUR LE POINT 'A'."
440 DISP "N'OUBLIEZ PAS DE STOPPER LE CHARIOT EN BOUT DE COURSE."
450 DISP "L'ORDINATEUR CESSERA SEUL DE PRENDRE LES LECTURES."
460 !
470 ! ACQUISITION DES DONNEES
480 !
490 OUTPUT 722 ; "PRESET"
500 OUTPUT 722 ; "TERM REAR"
510 OUTPUT 722 ; "CHAN 1"
520 OUTPUT 722 ; "DCV 40,.1"
530 DISP
540 PAUSE
550 CLEAR
560 DISP "ESSAI EN COUR."
570 DISP
580 DISP "FAIRE 'PAUSE' QUAND LE CHARIOT PASSERA SUR LE POINT B."
590 DISP @ DISP "FAIRE 'CONT 700' APRES AVOIR PESE SUR 'PAUSE'."
600 ! BOUCLE D'ACQUISITION

```

```


610 !
620 T1=TIME
630 FOR I=1 TO 500
640 ENTER 722 ; X(I) ! INFORMATION QUI ARRIVE DU VOLTMETRE
650 T2=TIME
660 NEXT I
670 !
680 !
690 ! BOUCLE TRANSFORMER EN (mm)
700 CLEAR @ GCLEAR
710 DISP
720 DISP "UN MOMENT S.V.P"
730 DISP
740 DISP "JE CALCULE"
750 !
760 FOR H=1 TO I-1
770  $Y(H)=(X(H)-X(1))*K1*S1*-1$ 
780 NEXT H
790  $V1=D1/(T2-T1)$ 
795  $D9=D1/I$ 
800 CLEAR @ DISP "DESIREZ VOUS 'STORER' LES DONNEES"
810 DISP "O/N"
820 INPUT R$
830 IF R$="O" THEN 840 ELSE 980
840 DISP "LE FICHIER PORTERE LE NO D'ESSAI" @ DISP
850 ! BOUCLE POUR LE STORAGE
860 ! DANS UN FICHIER PROTANT
870 ! AYANT POUR NOM LE NUMERO
880 ! DE L'ESSAI
890 !
900 CREATE F$,501,8 ! CREATION DU FICHIER DE DATA
910 ASSIGN# 1 TO F$ ! OUVERTURE DU FICHIER DE DATA
920 PRINT# 1 ; D9,I
930 FOR H=1 TO I-1
940 PRINT# 1 ; Y(H) ! ECRITURE DANS LE FICHIER DATA
950 NEXT H
960 ASSIGN# 1 TO * ! FERMETURE DU FICHIER DE DATA
980 DISP "DESIREZ VOUS REFAIRE L'ESSAI"
990 DISP "O/N"
1000 INPUT R$
1010 IF R$="O" THEN 10 ELSE 1020
1020 DISP "FIN"
9999 END

```

```

1 ! 88/09/30
2 ! NOM DU PROGRAMME: GRAPH5
3 ! -----
4 !
10 ! PROGRAMME DE TRAITEMENT POUR MESURES D'ORNIERES.
20 ! DONNEES POUR CALCULS
30 ! -----
40 ! SET-UP
50 ! -----
60 CLEAR
70 DIM X(500),Y(500),H(500)
80 MAT X=(0)
90 MAT Y=(0)
100 MAT H=(0)
110 ! -----
120 ! LECTURES
130 ! -----
140 CLEAR
150 DISP "NOM DU FICHIER A TRAITER"
160 INPUT F$
165 DISP
166 DISP "UN MOMENT, JE CALCULE."
170 ASSIGN# 1 TO F$
180 READ# 1 ; D9,I
190 FOR H=1 TO I-1
200 READ# 1 ; Y(H)
210 NEXT H
220 ASSIGN# 1 TO *
230 X(1)=0
240 FOR H=2 TO I-1
250 X(H)=D9*(H-1)
251 ! IF H=2 THEN Y(H)=(Y(H-1)+Y(H)+Y(H+1)+Y(H+2)+Y(H+3)+Y(H+4)+Y(H+5))/
7
252 ! IF H=3 THEN Y(H)=(Y(H-2)+Y(H-1)+Y(H)+Y(H+1)+Y(H+2)+Y(H+3)+Y(H+4)+Y
(H+5))/8
253 ! IF H=4 THEN Y(H)=(Y(H-3)+Y(H-2)+Y(H-1)+Y(H)+Y(H+1)+Y(H+2)+Y(H+3)+Y
(H+4)+Y(H+5))/9
255 ! IF H=5 THEN Y(H)=(Y(H-4)+Y(H-3)+Y(H-2)+Y(H-1)+Y(H)+Y(H+1)+Y(H+2)+Y
(H+3)+Y(H+4)+Y(H+5))/10
256 ! IF H=6 THEN Y(H)=(Y(H-5)+Y(H-4)+Y(H-3)+Y(H-2)+Y(H-1)+Y(H)+Y(H+1)+Y
(H+2)+Y(H+3)+Y(H+4))/11
257 ! IF H=6 THEN Y(H)=(Y(H)+Y(H+5))/11
258 IF H=2 THEN Y(H)=(Y(H-2)+Y(H-1)+Y(H)+Y(H+1)+Y(H+2))/5
260 NEXT H
300 DISP "ROUTE"
305 INPUT A$
310 DISP "TRONCON"
315 INPUT B$
320 DISP "SECTION"
325 INPUT C$
330 DISP "DATE"
335 INPUT D$
340 DISP "CHAINAGE"
345 INPUT E$
530 ! -----
540 ! TRACE DU GRAPHIQUE
550 ! -----
551 CLEAR
552 DISP "PREPARER LE TRACEUR DE COURBE ET FAIRE 'CONT'"
553 PAUSE

```

a modifier: lissage de courbe non valide


```

560 GCLEAR
570 PLOTTER IS 705
580 CSIZE 5,.5,0
590 DEG
600 FRAME
610 LOCATE 11.1,122.4,5.5,94.7
620 SCALE 0,4000,-80,80
630 XAXIS 0,500,0,4000
640 YAXIS 0,10,-70,10
650 LDIR 90
660 FOR X=500 TO 4000 STEP 500
670 MOVE X,15
680 LORG 5
690 LABEL VAL$(X)
700 NEXT X
710 LDIR 0
720 FOR Y=-70 TO 10 STEP 10
730 MOVE -145,Y
740 LORG 5
750 LABEL VAL$(Y)
760 NEXT Y
770 PENUP
780 MOVE 0,0
790 FOR I=1 TO H-5
800 PLOT X(I),Y(I)
810 NEXT I
820 MOVE 4100,0
830 LORG 1
840 CSIZE 5,.5,0
850 LABEL "MM"
860 MOVE 0,12
870 LORG 1
880 LABEL "MM"
890 CSIZE 8,.5,0
892 CLEAR
893 DISP "LE GRAPHIQUE EST-IL 'OK' (O/N)"
894 INPUT K$
895 IF K$="N" THEN 551 ELSE 900
900 MOVE 1000,-85
910 LABEL "MESURES D'ORNIERES"
920 MOVE 0,80
930 CSIZE 3,1,0
940 LABEL "ESSAI No: ";F$
950 MOVE 0,75
960 LABEL "ROUTE: ";A$
970 MOVE 0,70
980 LABEL "TRONCON: ";B$
990 MOVE 0,65
1000 LABEL "SECTION: ";C$
1010 MOVE 0,60
1020 LABEL "DATE: ";D$
1030 MOVE 0,55
1040 LABEL "CHAINAGE: ";E$
1050 MOVE 1700,80
1060 LABEL "GAUCHE"
1070 MOVE 3300,80
1080 LABEL "DROITE"
1090 DISP "VALEURS EN MEMOIRE"
1100 ! -----
1110 ! CALCUL DES MESURES

```

```

1120 ! -----
1130 M0=AMIN(Y)
1140 X6=0 @ Y6=0
1150 I=AMINROW
1160 ! -----
1170 ! RECHERCHE DE LA LECTURE LA PLUS PRES DE LA LARGEUR A MESURER
1180 ! -----
1190 CLEAR
1200 DISP "SUR QUELLE LARGEUR DE CHAUSSEE VOULEZ-VOUS CALCULER LES MESUR
ES?"
1210 DISP "INDIQUER EN mm."
1215 DISP "< OU = A: ";X(H-5); "mm"
1220 INPUT C3
1230 FOR I=H-1 TO 1 STEP -1
1240 IF X(I)<C3 THEN D1=X(I) @ GOTO 1260
1250 NEXT I
1260 D2=X(I+1)
1270 D3=D2-C3 @ D4=C3-D1
1280 IF D3>D4 THEN X3=X(I) @ N3=I ELSE X3=X(I+1) @ N3=I+1
1290 IF D3>D4 THEN Y3=Y(I) ELSE Y3=Y(I+1)
1300 ! X3,Y3 EST LE POINT LE PLUS PRES DE LA LONGEUR A MESURER
1310 T=ABS(100*Y3/X3)
1315 F3=X3 @ F4=Y3
1320 ! -----
1330 ! CALCULS DES POINTS D'APPUI DE LA REGLE POUR PRONFONDEUR D'ORNIERE
S
1340 ! -----
1350 CLEAR @ DISP "LONGEUR DE LA REGLE POUR MESURE D'ORNIERES (mm)";
1360 INPUT C1
1365 CLEAR
1366 DISP "PATIENCE, LES CALCULS SONT LONGS."
1370 FOR I=1 TO H-1
1380 IF SQR((X(I)-X(0))^2+(Y(I)-Y(0))^2)>=C1 THEN 1400 ELSE 1390
1390 NEXT I
1400 D1=SQR((X(I)-X6)^2+(Y(I)-Y6)^2)-C1
1410 D2=C1-SQR((X(I-1)-X6)^2+(Y(I-1)-Y6)^2)
1420 IF D2>D1 THEN X1=X(I) @ N1=I ELSE X1=X(I-1) @ N1=I-1
1430 IF D2>D1 THEN Y1=Y(I) ELSE Y1=Y(I-1)
1440 MAT H=(-100)
1450 A=Y6-Y1
1460 B=X1-X6
1470 C=-((Y6-Y1)*X6-(X6-X1)*Y6)
1480 FOR I=0 TO N1/2
1490 H(I)=(A*X(I)+B*Y(I)+C)/SQR(A^2+B^2)
1500 NEXT I
1510 D=AMAX(H)
1520 I=AMAXROW
1530 X6=X(I) @ Y6=Y(I) @ I6=I
1540 CLEAR
1550 DISP "REGLE UTILISEE: ";C1; " mm"
1560 DISP "ORNIERE GAUCHE"
1570 DISP
1580 DISP "X6: ";X6,"Y6: ";Y6
1590 MAT H=(-100)
1600 A=Y6-Y1
1610 B=X1-X6
1620 C=-((Y6-Y1)*X6-(X6-X1)*Y6)
1630 FOR I=N1/2 TO N1
1640 H(I)=(A*X(I)+B*Y(I)+C)/SQR(A^2+B^2)
1650 NEXT I

```

```

1660 D=AMAX(H)
1670 I=AMAXROW
1680 X1=X(I) @ Y1=Y(I) @ I1=I
1690 DISP "X1: ";X1,"Y1: ";Y1
1700 ! EQUATION DE LA DROITE ENTRE X6,Y6 ET X1,Y1 SOUS AX+BY+C=0
1710 ! -----
1720 M=(Y6-Y1)/(X6-X1) ! M=PENTE
1730 A=Y6-Y1
1740 B=X1-X6
1750 C=-((Y6-Y1)*X6-(X6-X1)*Y6)
1760 ! -----
1770 ! CALCUL DE LA PROFONDEUR DE L'ORNIERE
1780 ! -----
1790 MAT H=(0)
1800 H1=0
1810 FOR I=I6 TO I1
1820 H(I)=(A*X(I)+B*Y(I)+C)/SQR(A^2+B^2)
1830 NEXT I
1840 H1=AMIN(H)
1850 H1=ABS(H1) ! PROFONDEUR DE L'ORNIERE DE GAUCHE
1860 P1=H1
1870 DISP @ DISP "PROFONDEUR: ";H1
1880 ! -----
1890 ! CALCUL DE LA PROFONDEUR D'EAU DANS L'ORNIERE
1900 ! -----
1910 !
1920 G=0
1930 FOR I=I6 TO I1
1940 E=Y1-Y(I)
1950 IF G<E THEN G=E ELSE 1960
1960 NEXT I
1970 DISP @ DISP "PROFONDEUR D'EAU: ";G
1980 E1=G
1990 ! -----
2000 ! CALCUL POUR ORNIERE DE DROITE
2010 ! -----
2020 FOR I=H-1 TO 1 STEP -1
2030 IF SQR((X3-X(I))^2+(Y3-Y(I))^2)>=C1 THEN 2050 ELSE 2040
2040 NEXT I
2050 D1=SQR((X(I)-X3)^2+(Y(I)-Y3)^2)-C1
2060 D2=C1-SQR((X(I+1)-X3)^2+(Y(I+1)-Y3)^2)
2070 IF D2>D1 THEN X2=X(I) @ N2=I ELSE X2=X(I-1) @ N2=I-1
2080 IF D2>D1 THEN Y2=Y(I) ELSE Y2=Y(I-1)
2090 ! EQUATION DE LA DROITE ENTRE X2,Y2 ET X3,Y3 SOUS AX+BY+C=0
2100 ! -----
2110 A=Y2-Y3
2120 B=X3-X2
2130 C=-((Y2-Y3)*X2-(X2-X3)*Y2)
2140 DISP
2150 MAT H=(-100)
2160 FOR I=N2 TO N2+(N3-N2)/2
2170 H(I)=(A*X(I)+B*Y(I)+C)/SQR(A^2+B^2)
2190 NEXT I
2200 D=AMAX(H)
2210 I=AMAXROW
2220 X2=X(I) @ Y2=Y(I) @ I2=I
2230 DISP "ORNIERE DROITE"
2240 DISP
2250 DISP "X2: ";X2,"Y2: ";Y2
2260 MAT H=(-100)

```

```

2270 A=Y2-Y3
2280 B=X3-X2
2290 C=-((Y2-Y3)*X2-(X2-X3)*Y2)
2300 FOR I=N2+(N3-N2)/2 TO N3
2310 H(I)=(A*X(I)+B*Y(I)+C)/SQR(A^2+B^2)
2320 NEXT I
2330 D=AMAX(H)
2340 I=AMAXROW
2350 X3=X(I) @ Y3=Y(I) @ I3=I
2360 DISP "X3: ";X3,"Y3: ";Y3
2370 ! -----
2380 ! CALCUL DE LA PROFONDEUR DE L'ORNIERE
2390 ! -----
2391 ! EQUATION DE LA DROITE ENTRE X2,Y2 ET X3,Y3 SOUS AX+BY+C=0
2392 ! -----
2393 M=(Y2-Y3)/(X2-X3) ! M=PENTE
2394 A=Y2-Y3
2395 B=X3-X2
2396 C=-((Y2-Y3)*X2-(X2-X3)*Y2)
2397 ! -----
2400 MAT H=(0)
2410 H1=0
2420 FOR I=I2 TO I3
2430 H(I)=(A*X(I)+B*Y(I)+C)/SQR(A^2+B^2)
2440 NEXT I
2450 H1=AMIN(H)
2460 H1=ABS(H1) ! PROFONDEUR DE L'ORNIERE DE DROITE
2470 DISP @ DISP "PROFONDEUR: ";H1
2480 P2=H1
2490 ! -----
2500 ! CALCUL DE LA PROFONDEUR D'EAU DANS L'ORNIERE
2510 ! -----
2520 !
2530 G=0
2540 FOR I=I2 TO I3
2550 E=Y3-Y(I)
2560 IF G<E THEN G=E ELSE 2578
2578 NEXT I
2580 DISP @ DISP "PROFONDEUR D'EAU: ";G
2590 E2=G
2600 ! -----
2610 ! MESURE DES LARGEURS D'ORNIERES
2620 ! -----
2630 L1=SQR((X1-X6)^2+(Y6-Y1)^2)
2640 L2=SQR((X2-X3)^2+(Y2-Y3)^2)
2650 DISP
2660 DISP "L1: ";L1
2670 DISP "L2: ";L2
2680 ! -----
2690 ! SURFACES DES ORNIERES
2700 ! -----
2710 S1=0
2720 FOR I=I6 TO I1
2730 S1=S1+(X(I+1)-X(I))*((Y(I)+Y(I+1))/2)
2740 NEXT I
2750 S1=-S1
2760 S2=(Y6-Y1)*(X1-X6)/2
2770 S3=(X1-X6)*Y6
2780 S3=-S3
2790 F1=S1-S2-S3

```

```

2800 DISP "S1:";F1
2810 S1=0
2820 FOR I=I2 TO I3
2830 S1=S1+(X(I+1)-X(I))*((Y(I)+Y(I+1))/2)
2840 NEXT I
2850 S1=-S1
2860 S2=(Y2-Y3)*(X3-X2)/2
2870 S3=(X3-X2)*Y2
2880 S3=-S3
2890 F2=S1-S2-S3
2900 DISP "S2:";F2
2910 ! -----
2920 ! RESULTATS SUR LE GRAPHIQUE
2930 ! -----
2940 MOVE 1500,70
2950 LABEL USING 3470 ; "PROF.:";P1;"mm"
2960 MOVE 3050,70
2970 LABEL USING 3470 ; "PROF.:";P2;"mm"
2980 MOVE 1500,65
2990 LABEL USING 3470 ; "EAU:";E1;"mm"
3000 MOVE 3050,65
3010 LABEL USING 3470 ; "EAU:";E2;"mm"
3020 MOVE 1500,60
3030 LABEL USING 3480 ; "A-B:";L1;"mm"
3040 MOVE 3050,60
3050 LABEL USING 3480 ; "C-D:";L2;"mm"
3060 MOVE 1500,55
3070 LABEL USING 3480 ; "AIRE:";F1;"mm"
3080 MOVE 3050,55
3090 LABEL USING 3480 ; "AIRE:";F2;"mm"
3100 MOVE 2650,57
3110 CSIZE 2,1,0
3120 LABEL "2"
3130 MOVE 4185,57
3140 LABEL "2"
3150 CSIZE 3,1,0
3160 MOVE 0,50
3170 LABEL "REGLE:";C1;"mm"
3180 MOVE 0,45
3190 LABEL USING 3490 ; "DEVERS:";T;"/100"
3200 MOVE X6,Y6
3210 LORG 5
3220 CSIZE 3,1,0
3230 LABEL "X"
3240 MOVE X6,Y6-5
3260 LABEL "A"
3270 MOVE X1,Y1
3290 LABEL "X"
3300 MOVE X1,Y1-5
3320 LABEL "B"
3330 MOVE X2,Y2
3350 LABEL "X"
3360 MOVE X2,Y2+5
3380 LABEL "C"
3390 MOVE X3,Y3
3410 LABEL "X"
3420 MOVE X3,Y3-5
3440 LABEL "D"
3441 MOVE 0,0
3442 LABEL "*"

```

3443 MOVE F3,F4
3444 LABEL "*"
3450 DISP "TERMINE"
3460 END
3470 IMAGE 6A,DDZ.D,5A
3480 IMAGE 6A,DDDDD,8A
3490 IMAGE 7A,DZ.DD,5A

MINISTÈRE DES TRANSPORTS



QTR A 230 450